

8. Legkisebb négyzetes illesztés adatokra


Ha egy függvény közelítő értékét akarjuk meghatározni valamilyen előírt pontban, akkor alkalmazhatjuk az interpolációt, vagyis a függvényt az ismert pontjain áthaladó polinommal közelíthetjük. Az interpoláció alkalmazásánál feltételeztük, hogy a közelítés hibájának fő forrása az interpolációs formula képlethibája, nem pedig az ismert függvényértékekben levő kerekítési vagy mérési hiba.

A műszaki-gazdasági életben lejátszódó folyamatok leírása során konkrét mérésekből nyert információk alapján akarjuk meghatározni a keresett függvényt. A mért adatok pontosságára „nem lehet mérget venni”. Megpróbálunk olyan közelítő függvényt keresni, amely „simítja” a mérési eredményeket, azaz megpróbálja kiegyenlíteni a hibákat.

A legkisebb négyzetek módszerét először a csillagászatban használták, üstökösök pályájának leírásához. A módszer másik elnevezése, a regressziószámítás az angol *regression to the mean* (visszatérés az átlaghoz) kifejezésből származik. A módszer a statisztikai kiértékeléseknél is gyakran használatos.

A feladatot 2 lépésben oldjuk meg:

1.) Megállapítjuk az általános formulát, azaz a közelítő függvényt.



Ehhez célszerű a megfigyelt adatokat koordináta-rendszerben ábrázolni, s megrajzolni az adott pontok közé jól illeszkedő görbét. Így azonnal kitűnnek azok az adatok, amelyekben valószínűleg több a mérési hiba. A görbe megrajzolásához általános megfontolásokat is figyelembe kell venni, például azt, hogyan kell viselkednie a görbének 0-hoz közeli x esetén, nagy x értékekre, átmegy-e az origón, metszi-e a tengelyeket, stb. Ezek után a megrajzolt görbét összehasonlítjuk a legfontosabb elemi függvényekkel, és így meg tudjuk állapítani a függvényt leíró képlet típusát.

2.) A tapasztalati adatok alapján meghatározzuk a képletben szereplő paraméterek értékeit.

▼ 8. 1. Egyenes illesztés legkisebb négyzetek módszerével (a regressziós


egyenes)

A gyakorlatban sokszor előfordul, hogy N darab $(x_i, y_i) \quad i=1..N$ pontra lineáris függvényt kell illeszteni. Az illesztés pontosságát az **eltérések négyzetösszegével** mérjük. Az illesztendő egyenest vegyük fel

$$f(x, a, b) = ax + b$$

alakban, ahol a és b értékeit keressük!

Írjuk fel az x_i helyen mért y_i és a $f(x_i, a, b)$ függvényértékek eltéréseinek négyzetösszegét. (Ha a négyzetekkel számolunk, kiküszöböljük, hogy a pozitív és negatív előjelű eltérések kiegyenlítsék egymást. Számolhatnánk abszolútértékekkel is, de az nagyon megnehezítené a számolást.)



$$F := \sum_{i=1}^N (b + a x_i - y_i)^2$$

Ez az eltérés az a, b paraméterek kétváltozós függvénye. Ennek a függvénynek a minimumát keressük, vagyis a minimumot adó

$$\frac{\partial}{\partial a} F = 0, \quad \frac{\partial}{\partial b} F = 0$$

egyenletrendszert kell megoldani. A deriválás elvégzése és átrendezés után a következő, a és b ismeretleneket tartalmazó lineáris egyenletrendszert kajuk:

$$b \left(\sum_{i=1}^N x_i \right) + a \left(\sum_{i=1}^N x_i^2 \right) = \sum_{i=1}^N y_i x_i,$$
$$b N + a \left(\sum_{i=1}^N x_i \right) = \sum_{i=1}^N y_i.$$

bevezetve az

$$\frac{\sum_{i=1}^N x_i}{N} = x_atl, \quad \frac{\sum_{i=1}^N y_i}{N} = y_atl$$

jelöléseket, az egyenletrendszer megoldása jól megjegyezhető alakot ölt.

$$a = \frac{\sum_{i=1}^N y_i x_i - N \cdot x_atl \cdot y_atl}{\sum_{i=1}^N x_i^2 - N x_atl^2}, \quad b = y_atl - a x_atl$$

▼ 8. 1. 1. Egyenesillesztés Maple-lel

Használt Maple parancsok: [LeastSquares](#)

1. Példa Adjuk meg az $X := [1, 1.5, 2, 2.5, 3, 3.5, 4]$, $Y := [5, 5.4, 7.8, 9.5, 11.1, 12.3, 15.2]$ halmazokkal adott pontokra a legkisebb négyzetes közelítést adó lineáris polinomot!

Megoldás

Az x_i és y_i adatokat egy X és egy Y vektorban adjuk meg. A *CurveFitting* csomag tartalmazza a legkisebb négyzetes illesztés *LeastSquare*(Független változó adatai, Függő változó adatai, független változó, curve=tipus) parancsát.

```
> restart : with(CurveFitting) : X := [1, 1.5, 2, 2.5, 3, 3.5, 4] :
Y := [5, 5.4, 7.8, 9.5, 11.1, 12.3, 15.2] : N := nops(X) :
> y := LeastSquares(X, Y, x, curve = b + a x)
y := 0.953571428571426 + 3.40714285714286 x
```

Számítsuk ki az egyes mért Y elemek eltérését az egyenes megfelelő függvényértékétől, illetve az eltérések négyzetösszegét!

```
> elteresek := [seq([ [X_p, Y_i], [X_p, subs(x=X_p, y)] ], i=1..N) ]
elteresek := [[ [1, 5], [1, 4.36071428571428]], [[1.5, 5.4], [1.5, 6.06428571428571]], [[2,
7.8], [2, 7.76785714285714]], [[2.5, 9.5], [2.5, 9.47142857142857]], [[3, 11.1], [3,
11.1750000000000]], [[3.5, 12.3], [3.5, 12.8785714285714]], [[4, 15.2], [4,
14.5821428571429]]]
```

```
> sum_{i=1}^N (subs(x=X_p, y) - Y_i)^2 = sum_{i=1}^N (subs(x=X_p, y) - Y_i)^2
sum_{i=1}^7 (0.953571428571426 + 3.40714285714286 X_i - Y_i)^2 = 1.573928570
```


Rajzoljuk fel a pontokat, az egyenest, és az eltéréseket.

```
> N := nops(X) :
pontok := plot([seq([X_p, Y_i], i=1..N)], style=point, symbol=diamond, color=black) :
x0 := min(seq(X_p, i=1..N)) : xv := max(seq(X_p, i=1..N)) : y0 := min(seq(Y_i, i=1..N)) : yv := max(seq(Y_i, i=1..N))
```

```

..N) ) : yv := max(seq(Yi, i = 1 ..N) ) : graf := plot(y, x = x0 - 1 ..xv, color = red) :
rajz := plot(elteresek, color = black, thickness = 3) : plots[display]([pontok, graf,
rajz], axes = framed);

```



▼ 8. 2. Magasabbfokú polinomok illesztése legkisebb négyzetes módszerrel

2. Példa Vegyük az 1. példa adatait, illesszünk a pontokra másodfokú görbét.

Megoldás

▼ 8. 2. 1. Megoldás lépésként

```

> restart : X := [ 1, 1.5, 2, 2.5, 3, 3.5, 4 ] :
Y := [ 5, 5.4, 7.8, 9.5, 11.1, 12.3, 15.2 ] : N := nops(X) :
> f := x → a x2 + b x + c :

```

Képezzük a mért és a számított értékek eltérések négyzetösszegét!

```

> F := simplify( sum( (f(Xi) - Yi)2, i = 1 .. N ) )
F := -132.6000000 c - 1203. a - 379.2000000 b + 548.1875000 a2 + 323.7500000 a b
+ 101.5000000 a c + 50.75000000 b2 + 35. b c + 7. c2 + 710.7900000

```

Az (a, b, c) ismeretlen paramétereket határozzuk meg úgy, hogy az eltérések összegének minimuma legyen ezen a helyen! Ehhez a parciális deriváltaknak el kell venni.

```

> e1 := ∂/∂a F = 0; e2 := ∂/∂b F = 0; e3 := ∂/∂c F = 0
e1 := -1203. + 1096.375000 a + 323.7500000 b + 101.5000000 c = 0

```

$$e2 := -379.2000000 + 323.7500000 a + 101.5000000 b + 35. c = 0$$

$$e3 := -132.6000000 + 101.5000000 a + 35. b + 14. c = 0$$

Egy 3×3-as lineáris egyenletrendszer kell megoldani.

```
> evalf(solve( {e1, e2, e3}, {a, b, c} ))
      {a=0.3000000000, b=1.907142857, c=2.528571429}
```

```
> parabola := subs(%, f(x))
      parabola := 0.3000000000 x^2 + 1.907142857 x + 2.528571429
```

▼ 8. 2. 2. Megoldás a CurveFitting[LeastSquare] parancsával


```
> with(CurveFitting) : y := LeastSquares(X, Y, x, curve = a x^2 + b x + c)
      y := 2.52857142857142 + 1.90714285714286 x + 0.299999999999999 x^2
```

```
> elteresek := [seq( [X[i], Y[i]], [X[i], subs(x=X[i], y)] ), i=1..N) ];
      elteresek := [[1, 5], [1, 4.73571428571428], [1.5, 5.4], [1.5, 6.06428571428571], [2,
      7.8], [2, 7.54285714285714], [2.5, 9.5], [2.5, 9.17142857142857], [3, 11.1], [3,
      10.9500000000000], [3.5, 12.3], [3.5, 12.8785714285714], [4, 15.2], [4,
      14.9571428571429]]]
```

$$\begin{aligned} > \sum_{i=1}^N (\text{subs}(x=X_i, y) - Y_i)^2 &= \sum_{i=1}^N (\text{subs}(x=X_i, y) - Y_i)^2 \\ &= \sum_{i=1}^7 (2.52857142857142 + 1.90714285714286 X_i + 0.299999999999999 X_i^2 - Y_i)^2 \\ &= 1.101428572 \end{aligned}$$

A négyzetes eltérések összege kisebb, mint az egyenestől való eltérés. Rajzoljuk fel a pontokat és a parabolát!

```
> N := nops(X) :
      pontok := plot( [seq( [X_p, Y_i], i=1..N) ], style=point, symbol=diamond, color=black) :
      x0 := min(seq(X_p, i=1..N)) : xv := max(seq(X_p, i=1..N)) : y0 := min(seq(Y_p, i=1..N)) :
      yv := max(seq(Y_p, i=1..N)) : graf := plot(y, x=x0-1..xv, color=red) :
      rajz := plot(elteresek, color=black, thickness=3) :
> plots[display]([pontok, graf, rajz], axes=framed)
```



▼ 8. 3. Linearizálás

A következő példa azt mutatja, hogy néha sem a lépésenkénti megoldás, sem a *LeastSquare* parancs használata nem vezet eredményre.

3. Példa A pszichológusok vizsgálták a felejtés folyamatát. Az emlékezetben tartás és a felejtés ellentétes folyamatok, a kettőt általában együtt szokták vizsgálni. A kísérleti vizsgálatok kimutatták, hogy a felejtés, bár annál nagyobb, minél hosszabb idő telt el az emlékezetbe vésés óta, jóval gyorsabb kezdetben, mint később. Kísérletek során az emlékezetben tartást vizsgálták: a megtanult anyagot 100%-nak véve, az emlékezetben tartott anyag az idő múlásával a következő módon változott

idő	20 perc	1 óra	9 óra	24 óra	2 nap	6 nap	31 nap
emlékezet	58 %	44%	34%	36%	28%	25%	21%

- a; Adjuk meg a felejtés (pontosabban az emlékezetben tartás) görbéjét!
- b; A megtanult anyag kb. hány százaléka marad meg 2 hét múlva?


Megoldás

Legyenek a X halmaz elemei az idő értékei órában, Y elemei pedig az emlékezet %-os értékei.

> *restart*; $X := \left[\frac{1}{3}, 1, 9, 24, 2 \cdot 24, 6 \cdot 24, 31 \cdot 24 \right]$; $Y := [58, 44, 34, 36, 28, 25, 21]$:

A közelítő függvény kiválasztásához rajzoljuk fel a megfelelő pontokat!

```
> N := nops(X) :
pontok := plot( [seq( [Xi, Yi], i = 1 .. N) ], style = point, symbol = cross, symbolsize = 20, color
= red, view = [1 .. 750, 0 .. 70]) :
pontok
```




A pontok nem valószínű, hogy lineáris kapcsolatot mutatnak. Próbálkozzunk először

$y = \frac{a}{x} + b$ típusú függvénnyel.

```
> with(CurveFitting) : y := evalf( LeastSquares( X, Y, x, curve =  $\frac{a}{x} + b$  ) )
y := 29.01478367 +  $\frac{10.25766230}{x}$ 
```

```
> n := nops(X) :
pontok := plot( [seq( [Xi, Yi], i = 1 .. n) ], style = point, symbol = diamond, symbolsize = 20,
color = black) :
x0 := min( seq( Xi, i = 1 .. n ) ) : xv := max( seq( Xi, i = 1 .. n ) ) : y0 := min( seq( Yi, i = 1 .. n ) ) :
yv := max( seq( Yi, i = 1 .. n ) ) : graf := plot( y, x = x0 .. xv, color = red) :
> plots[display]( [pontok, graf], view = [x0 .. xv, y0 - 1 .. yv], axes = framed)
```



Az ábrából látszik, hogy a pontokhoz tartozó legjobb hiperbola is meglehetősen pontatlan. Nézzük meg a négyzeteltéréseket!

$$> \sum_{i=1}^n (\text{subs}(x = 'X'_i, y) - 'Y'_i)^2 = \sum_{i=1}^n (\text{subs}(x = X_i, y) - Y_i)^2; \text{emlekezés} := \text{subs}(x = 14 \cdot 24, y) :$$

$$\sum_{i=1}^7 \left(29.01478367 + \frac{10.25766230}{X_i} - Y_i \right)^2 = 166.0011616$$

Ha hiperbolával közelítenénk, az emlékezés 2 hét (336 óra) múlva 29%, ami több, mint a 6 nap után mért érték. Próbálkozzunk valamilyen más, negatív kitevőjű hatványfüggvénnyel, legyen $y = a x^b$ alakú.

```
> y := LeastSquares(X, Y, x, curve = a x^b)
Error, (in CurveFitting:-LeastSquares) curve to fit is not linear in the parameters
```

Hibaüzenetet kapunk, a parancs erre a függvénytípusra nem végrehajtható. Nézzük meg, hátha lépésenként elvégezhető a feladat megoldása.

```
> f := x → a x^b :
```

A jól ismert módon képezzük az eltérések négyzetösszegét!

$$\begin{aligned}
 &> F := \text{simplify} \left(\sum_{i=1}^N (f(X_i) - Y_i)^2 \right) \\
 F &:= a^2 9^{-b} - 116 a 3^{-b} + 9602 + a^2 - 88 a + a^2 81^b - 68 a 9^b + a^2 576^b - 72 a 24^b + a^2 2304^b \\
 &\quad - 56 a 48^b + a^2 20736^b - 50 a 144^b + a^2 553536^b - 42 a 744^b
 \end{aligned}$$

Az (a, b) ismeretlen paramétereket határozzuk meg úgy, hogy az eltérések összegének minimuma legyen ezen a helyen. A parciális deriváltakat tegyük 0-val egyenlővé.

$$\begin{aligned}
 &> e1 := \frac{\partial}{\partial a} F = 0; e2 := \frac{\partial}{\partial b} F = 0 \\
 e1 &:= 2 a 9^{-b} - 116 3^{-b} + 2 a - 88 + 2 a 81^b - 68 9^b + 2 a 576^b - 72 24^b + 2 a 2304^b - 56 48^b \\
 &\quad + 2 a 20736^b - 50 144^b + 2 a 553536^b - 42 744^b = 0 \\
 e2 &:= -2 a^2 9^{-b} \ln(3) + 116 a 3^{-b} \ln(3) + 4 a^2 81^b \ln(3) - 136 a 9^b \ln(3) + 2 a^2 576^b \ln(24) \\
 &\quad - 72 a 24^b \ln(24) + 2 a^2 2304^b \ln(48) - 56 a 48^b \ln(48) + 4 a^2 20736^b \ln(12) \\
 &\quad - 100 a 144^b \ln(12) + 2 a^2 553536^b \ln(744) - 42 a 744^b \ln(744) = 0
 \end{aligned}$$

Látszik, hogy a -ra, b -re bonyolult, nem lineáris egyenletrendszeret kaptunk.

> solve({e1, e2}, {a, b})

Warning, computation interrupted

A megoldás nagyon hosszú ideig tart, ezért megszakítottuk. Más utat kell keresnünk. Alakítsuk át a keresett hatványfüggvényt lineáris függvénné úgy, hogy mindkét oldal természetes alapú logaritmusát vesszük.

> restart : with(CurveFitting) : linerizal := $\ln(y) = \ln(a x^b)$:

A logaritmus azonosságai miatt:

> linerizal := $\ln(y) = \ln(a) + b \cdot \ln(x)$:

Vezessünk be új változókat, legyen $\ln(y) = u$, $\ln(a) = A$, $\ln(x) = v$

> subs({ln(a) = A, ln(x) = v, ln(y) = u}, linerizal)
 $u = A + b v$

Így már lineáris kifejezést kapunk v , u adatokra, A és b értékeit egyenesillesztéssel megkaphatjuk. A X és Y halmaz elemeit is logaritmálni kell, hogy U és V halmazok elemeit megkapjuk.

> $X := \left[\frac{1}{3}, 1, 9, 24, 2 \cdot 24, 6 \cdot 24, 31 \cdot 24 \right]$: $Y := [58, 44, 34, 36, 28, 25, 21]$: $N := \text{nops}(X)$:

> $V := [\text{seq}(\text{evalf}(\ln(X_i)), i = 1 .. N)]$: $U := [\text{seq}(\text{evalf}(\ln(Y_i)), i = 1 .. N)]$

$V := [-1.098612289, 0., 2.197224578, 3.178053830, 3.871201011, 4.969813300, 6.612041035]$

$U := [4.060443011, 3.784189634, 3.526360525, 3.583518938, 3.332204510, 3.218875824, 3.044522438]$

> egyenes := $u = \text{LeastSquares}(V, U, v, \text{curve} = b v + A)$

$\text{egyenes} := u = 3.85951880498793 - 0.125015285151951 v$

Most alakítsuk vissza az egyenletet:

> *logaritmusos* := *subs*({*u* = $\ln(y)$, *v* = $\ln(x)$ }, *egyenest*)
logaritmusos := $\ln(y) = 3.85951880498793 - 0.125015285151951 \ln(x)$


Ebből kell *y*-t kifejezni, mindkét oldalt *e* hatványára kell emelni.

> *hatvany* := $e^{\text{lhs}(\text{logaritmusos})} = \text{expand}(e^{\text{rhs}(\text{logaritmusos})})$
hatvany := $y = \frac{47.4425167729163}{x^{0.125015285151951}}$

> $\sum_{i=1}^N (\text{subs}(x = 'X'_p, \text{rhs}(\text{hatvany})) - 'Y'_i)^2 = \sum_{i=1}^N (\text{subs}(x = X_p, \text{rhs}(\text{hatvany})) - Y_i)^2$
 $\sum_{i=1}^7 \left(\frac{47.4425167729163}{X_i^{0.125015285151951}} - Y_i \right)^2 = 47.55712871$

Az eltérések négyzetösszege már jóval kisebb, mint a hiperbolánál kapott 166.0011616 -os érték.

> *pontok* := *plot*([*seq*([*X_p*, *Y_i*], *i* = 1 .. *N*)], *style* = *point*, *symbol* = *diamond*, *symbolsize* = 20,
color = *black*) :
x0 := *min*(*seq*(*X_p*, *i* = 1 .. *N*)) : *xv* := *max*(*seq*(*X_p*, *i* = 1 .. *N*)) :
y0 := *min*(*seq*(*Y_i*, *i* = 1 .. *N*)) : *yv* := *max*(*seq*(*Y_i*, *i* = 1 .. *N*)) : *graf* := *plot*(*rhs*(*hatvany*), *x*
= *x0* .. *xv*, *color* = *red*) :
> *plots*[*display*]([*pontok*, *graf*], *view* = [1 .. 750, 0 .. 70], *axes* = *framed*)



Az emlékezés 2 hét (336 óra) múlva már sokkal pontosabb illesztésnek felel meg

> `emlekezés := subs(x = 14·24, rhs(hatvany));`
`emlekezés := 22.9264437799361`

Annak vizsgálatára, hogy melyik közelítés a jobb, a matematikai statisztika hibaanalizáló módszerei alkalmazandók.

▼ 8. 4. Illesztés tetszőleges függvénnyel

Nemcsak polinom illesztést végezhetünk el a fenti módszerrel! Tetszőleges függvényeket alapfüggvényként tekinthetünk és ezek lineáris kombinációját képezhetjük, majd ezt a függvényt tekintjük közelítő függvénynek.

4. Példa Illesszük az $y = \ln(x)$, $y = \cos(x)$, $y = e^x$ függvények lineáris kombinációját az

$X := [0.24, 0.65, 0.95, 1.24, 1.73, 2.01, 2.23, 2.52, 2.77, 2.99];$
 $Y := [0.23, -0.26, -1.1, -0.45, 0.27, 0.1, -0.29, 0.24, 0.56, 1]$
 adathalmazra.

Megoldás

Az $y = a \ln(x) + b \cos(x) + c e^x$ függvény együtthatóit a

$F(a, b, c) := \sum_{i=1}^N \left(a \ln(x_i) + b \cos(x_i) + c e^{x_i} - y_i \right)^2$ háromváltozós függvény a , b és c változója szerinti parciális deriváltjai zérushelyeinek megkeresésével adjuk meg.

> $f := x \rightarrow a \ln(x) + b \cos(x) + c e^x$;
 $X := [0.24, 0.65, 0.95, 1.24, 1.73, 2.01, 2.23, 2.52, 2.77, 2.99]$; $Y := [0.23, -0.26, -1.1, -0.45, 0.27, 0.1, -0.29, 0.24, 0.56, 1]$; $N := \text{nops}(X)$:

> $F := \text{simplify} \left(\sum_{i=1}^N (f(X_i) - Y_i)^2 \right)$

$F := -3.232530229 a + 4.765416705 b - 53.54553573 c + 3.071200000 + 5.108423881 b^2$
 $- 98.01717203 b c + 1002.506504 c^2 + 6.794102805 a^2 - 10.69498697 a b$
 $+ 126.5177848 a c$

> $e1 := \frac{\partial}{\partial a} F = 0$; $e2 := \frac{\partial}{\partial b} F = 0$; $e3 := \frac{\partial}{\partial c} F = 0$
 $e1 := -3.232530229 + 13.58820561 a - 10.69498697 b + 126.5177848 c = 0$
 $e2 := 4.765416705 + 10.21684776 b - 98.01717203 c - 10.69498697 a = 0$
 $e3 := -53.54553573 - 98.01717203 b + 2005.013008 c + 126.5177848 a = 0$

Lineáris egyenletrendszer kaptunk, ami az ismert módszerek valamelyikével megoldható.


> $\text{solve}(\{e1, e2, e3\}, \{a, b, c\})$
 $\{a = -1.041032215, b = -1.261318782, c = 0.03073482571\}$

Adjuk meg a közelítő függvényt!

> $\text{kozelito} := \text{subs}(\%, f(x))$
 $\text{kozelito} := -1.041032215 \ln(x) - 1.261318782 \cos(x) + 0.03073482571 e^x$

Rajzoljuk fel a pontokat és a kapott közelítő függvényt!

> $N := \text{nops}(X)$:
 $\text{pontok} := \text{plot}([seq([X_i, Y_i], i = 1 .. N)], \text{style} = \text{point}, \text{symbol} = \text{diamond}, \text{color} = \text{black})$:
 $\text{graf} := \text{plot}(\text{kozelito}, x = X_1 .. X_N, \text{color} = \text{red})$:
> $\text{plots}[\text{display}]([\text{pontok}, \text{graf}], \text{view} = [0 .. 3, -1 .. 1], \text{scaling} = \text{constrained}, \text{title} = \text{Legkisebb négyzetes illesztés 3 alapfüggvénnyel}, \text{axes} = \text{framed})$



Az eltérések négyzetösszege

> $\text{Sum}((\text{subs}(x=X[k], \text{kozelito}) - Y[k])^2, k=1 \dots N) = \text{sum}((\text{subs}(x=X[k], \text{kozelito}) - Y[k])^2, k=1 \dots N);$

$$\sum_{k=1}^{10} \left(-1.041032215 \ln(X_k) - 1.261318782 \cos(X_k) + 0.03073482571 e^{X_k} - Y_k \right)^2 = 0.9255728970$$

▼ 8. 5. A regressziós együtthatók meghatározása a lineáris algebra eszközeivel

▼ 8.5.1. Egyváltozós regresszió

Határozzuk meg a lineáris regresszió együtthatóit mátrixok segítségével is. Az egyenletrendszer együtthatómátrixa legyen **A**, konstansokból álló mátrixa pedig **B**.

> $\text{restart}; \text{with}(\text{LinearAlgebra}) : e1 := b \left(\sum_{i=1}^N x_i \right) + a \left(\sum_{i=1}^N x_i^2 \right) = \sum_{i=1}^N y_i x_i : e2 := b N$

$$+ a \left(\sum_{i=1}^N x_i \right) = \sum_{i=1}^N y_i :$$

> $A, B := \text{GenerateMatrix}([e1, e2], [a, b]);$

$$A, B := \begin{bmatrix} \sum_{i=1}^N x_i^2 & \sum_{i=1}^N x_i \\ \sum_{i=1}^N x_i & N \end{bmatrix}, \begin{bmatrix} \sum_{i=1}^N y_i x_i \\ \sum_{i=1}^N y_i \end{bmatrix}$$

Az **A** mátrix előállítás az x adatok alapján, $N=3$ esetre:

> $N := 3; A; B$

$N := 3$

$$\begin{bmatrix} x_1^2 + x_2^2 + x_3^2 & x_1 + x_2 + x_3 \\ x_1 + x_2 + x_3 & 3 \end{bmatrix}$$

$$\begin{bmatrix} y_1 x_1 + y_2 x_2 + y_3 x_3 \\ y_1 + y_2 + y_3 \end{bmatrix}$$

A megoldást az inverz mátrix segítségével adjuk meg:

> $Megoldas := simplify(MatrixInverse(A).B);$

$$Megoldas := \left[\left[\frac{1}{2} \frac{2 y_1 x_1 + 2 y_2 x_2 + 2 y_3 x_3 - x_1 y_2 - x_1 y_3 - x_2 y_1 - y_3 x_2 - x_3 y_1 - y_2 x_3}{x_1^2 + x_2^2 + x_3^2 - x_1 x_2 - x_1 x_3 - x_2 x_3} \right], \right.$$

$$\left[\frac{1}{2} \frac{1}{x_1^2 + x_2^2 + x_3^2 - x_1 x_2 - x_1 x_3 - x_2 x_3} \left(-x_1 y_2 x_2 - x_1 y_3 x_3 - x_2 y_1 x_1 - x_2 y_3 x_3 \right. \right.$$

$$\left. \left. - x_3 y_1 x_1 - x_3 y_2 x_2 + x_1^2 y_2 + x_1^2 y_3 + x_2^2 y_1 + x_2^2 y_3 + x_3^2 y_1 + x_3^2 y_2 \right) \right] \right]$$

Ellenőrizzük a megoldást, úgy hogy az $e1$, $e2$ egyenletrendszerben a -ba a $Megoldas$ első elemét, b -be a második elemét helyettesítjük.

> $simplify(subs(\{a = Megoldas_1, b = Megoldas_2\}, \{e1, e2\}))$
 $\{y_1 + y_2 + y_3 = y_1 + y_2 + y_3, y_1 x_1 + y_2 x_2 + y_3 x_3 = y_1 x_1 + y_2 x_2 + y_3 x_3\}$

Most próbáljuk meg az **A** mátrixot jól megjegyezhető, egyszerűbb módon is előállítani. Legyenek a független változó adatai X , a függő változó értékei a Y halmazban.

> $X := Matrix(1, 3, [x_1, x_2, x_3]) : Y := Matrix(1, 3, [y_1, y_2, y_3]) :$

Hozzunk létre egy olyan mátrixot, amelynek első sorában az X elemei, második sorában csupa 1-es áll. A *Matrix* parancs *fill=érték* opciója a mátrix nem definiált elemi helyére ír 1 értéket.

> $XA := Matrix(2, 3, X, fill = 1)$

$$XA := \begin{bmatrix} x_1 & x_2 & x_3 \\ 1 & 1 & 1 \end{bmatrix}$$

Képezzük az így kapott mátrix és transzponáltjának szorzatát, és bizonyítsuk be, hogy ez éppen a **A** mátrixot adja!

> $XA.Transpose(XA); A = XA.Transpose(XA)$

$$\begin{bmatrix} x_1^2 + x_2^2 + x_3^2 & x_1 + x_2 + x_3 \\ x_1 + x_2 + x_3 & 3 \end{bmatrix} = \begin{bmatrix} x_1^2 + x_2^2 + x_3^2 & x_1 + x_2 + x_3 \\ x_1 + x_2 + x_3 & 3 \end{bmatrix}$$

Most képezzük az **XA** és az **Y** transzponáltjának szorzatát, lássuk be, hogy ez éppen a **B** mátrix.

> $XA.Transpose(Y); B = XA.Transpose(Y)$

$$\begin{bmatrix} y_1 x_1 + y_2 x_2 + y_3 x_3 \\ y_1 + y_2 + y_3 \end{bmatrix} = \begin{bmatrix} y_1 x_1 + y_2 x_2 + y_3 x_3 \\ y_1 + y_2 + y_3 \end{bmatrix}$$

Vagyis az **A** mátrix előállítható az **XA** $2 \times N$ -es mátrix és transzponáltjának szorzataként, ahol az első sorban az N darab adat első koordinátája, a második sorban csupa egyes áll, míg a **B** mátrix az **XA** és az **Y** transzponáltjának szorzataként áll elő. Tehát az együtthatók, vagyis a megoldásmátrix

$$(\mathbf{XA} \cdot \mathbf{XA}^T)^{-1} \cdot \mathbf{XA} \cdot \mathbf{Y}^T$$

első eleme az a , második eleme b értékét adta.

Tehát ha a regressziós egyenes együtthatóit keressük, nem kell mást tennünk, mint az **XA** és **Y** mátrixokon elvégezni a $(\mathbf{XA} \cdot \mathbf{XA}^T)^{-1} \cdot \mathbf{XA} \cdot \mathbf{Y}^T$ mátrixműveleteket.

▼ 8.5.2. Többváltozós regresszió

Többváltozós regresszióról akkor beszélünk, ha az eredményt több változó befolyásolja.

Vizsgáljuk először a $n=2$ kétváltozós lineáris regressziót! Bizonyítsuk be, hogy a megoldás az egyváltozós esethez hasonló, csak az **XA** mátrixnak 3 sora van, ami az $X1$, $X2$ adatsorokat és egy csupa 1-esből álló sort tartalmaz.

A független változók legyenek $x1$ és $x2$, a keresett kétváltozós függvény legyen $f(x1, x2) = a0 + a1 \cdot x1 + a2 \cdot x2$ alakú (vagyis síkot illesztünk az adatokra.) . Az $a0$, $a1$, $a2$ ismeretleneket tartalmazó háromváltozós függvény pedig

$$F(a0, a1, a2) = \sum_{i=1}^N (a0 + a1 x1_i + a2 x2_i - y_i)^2 \text{ alakú.}$$

> *restart; with(CurveFitting) :*

$$\text{with(LinearAlgebra)} : F := \sum_{i=1}^N (a0 + a1 x1_i + a2 x2_i - y_i)^2 : e1 := \frac{\partial}{\partial a2} F = 0 : e2$$

$$\begin{aligned}
& := \frac{\partial}{\partial a1} F = 0 : e3 := \frac{\partial}{\partial a0} F = 0 : \\
> \quad e1 &:= a0 \left(\sum_{i=1}^N x2_i \right) + a1 \left(\sum_{i=1}^N x1_i x2_i \right) + a2 \left(\sum_{i=1}^N x2_i^2 \right) = \sum_{i=1}^N y_i x2_i : \\
> \quad e2 &:= a0 \left(\sum_{i=1}^N x1_i \right) + a1 \left(\sum_{i=1}^N x1_i^2 \right) + a2 \left(\sum_{i=1}^N x1_i x2_i \right) = \sum_{i=1}^N y_i x1_i : \\
> \quad e3 &:= a0 N + a1 \left(\sum_{i=1}^N x1_i \right) + a2 \left(\sum_{i=1}^N x2_i \right) = \sum_{i=1}^N y_i :
\end{aligned}$$

Amint azt vártuk, a három változóra lineáris egyenletrendszeret kaptunk.

> $A, B := \text{GenerateMatrix}([e1, e2, e3], [a2, a1, a0])$

$$A, B := \begin{bmatrix} \sum_{i=1}^N x2_i^2 & \sum_{i=1}^N x1_i x2_i & \sum_{i=1}^N x2_i \\ \sum_{i=1}^N x1_i x2_i & \sum_{i=1}^N x1_i^2 & \sum_{i=1}^N x1_i \\ \sum_{i=1}^N x2_i & \sum_{i=1}^N x1_i & N \end{bmatrix}, \begin{bmatrix} \sum_{i=1}^N y_i x2_i \\ \sum_{i=1}^N y_i x1_i \\ \sum_{i=1}^N y_i \end{bmatrix}$$

> $\text{Megoldas} := \text{simplify}(\text{MatrixInverse}(A).B) :$

A fenti mátrix áttekinthetetlenül nagy, ezért ki sem írtuk. Írjuk fel a megoldást $(\mathbf{XA} \cdot \mathbf{XA}^T)^{-1} \cdot \mathbf{XA} \cdot \mathbf{Y}^T$ alakban, legyen $n=2, N=5$.

> $X1 := [x1_1, x1_2, x1_3, x1_4, x1_5] : X2 := [x2_1, x2_2, x2_3, x2_4, x2_5] : Y := [y_1, y_2, y_3, y_4, y_5] :$
 $N := \text{nops}(X1) :$

> $A := \text{Matrix}(3, 3, [[\text{evalf}(A_{1,1}), \text{evalf}(A_{1,2}), \text{evalf}(A_{1,3})], [\text{evalf}(A_{2,1}), \text{evalf}(A_{2,2}), \text{evalf}(A_{2,3})], [\text{evalf}(A_{3,1}), \text{evalf}(A_{3,2}), \text{evalf}(A_{3,3})]]) :$

> $B := \text{Matrix}(3, 1, [\text{evalf}(B_1), \text{evalf}(B_2), \text{evalf}(B_3)]) :$

Az $\mathbf{XA}$ mátrixban most az $X1, X2$ adatai kerülnek az első 2 sorba, a harmadik sorba kerül csupa 1-es. Igazoljuk, $\mathbf{A} = \mathbf{XA} \cdot \mathbf{XA}^T$ és $\mathbf{B} = \mathbf{XA} \cdot \mathbf{Y}^T$.

> $XA := \text{Matrix}(3, N, [X2, X1], \text{fill} = 1)$

$$XA := \begin{bmatrix} x2_1 & x2_2 & x2_3 & x2_4 & x2_5 \\ x1_1 & x1_2 & x1_3 & x1_4 & x1_5 \\ 1 & 1 & 1 & 1 & 1 \end{bmatrix}$$

> $\mathbf{A} = (\mathbf{XA}.\text{Transpose}(\mathbf{XA}))$

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

> $B - XA.Transpose(convert(Y, Matrix))$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Megjegyzés: mivel Y halmaz és nem mátrix, át kell konvertálni mátrixszá. Tehát a megoldás:

> $Megoldas := simplify(MatrixInverse(XA.Transpose(XA)) . XA.Transpose(convert(Y, Matrix)))$:

Ez persze így is sok számolást igényel, de könnyebben megjegyezhető a számolás algoritmus.

5. Példa Írjuk fel a kétváltozós regressziós függvényt, ha $X1 := [58, 52, 133, 179, 98]$, $X2 := [405, 450, 350, 285, 330]$, $Y := [590, 660, 780, 770, 710]$.

Megoldás

> $X1 := [58, 52, 133, 179, 98]$; $X2 := [405, 450, 350, 285, 330]$;
 $Y := [590, 660, 780, 770, 710]$; $N := nops(Y)$:

> $XA := Matrix(3, N, [X2, X1], fill = 1)$

$$XA := \begin{bmatrix} 405 & 450 & 350 & 285 & 330 \\ 58 & 52 & 133 & 179 & 98 \\ 1 & 1 & 1 & 1 & 1 \end{bmatrix}$$

> $XA.Transpose(XA)$

$$\begin{bmatrix} 679150 & 176795 & 1820 \\ 176795 & 65402 & 520 \\ 1820 & 520 & 5 \end{bmatrix}$$

> $XA.Transpose(convert(Y, Matrix))$

$$\begin{bmatrix} 1262700 \\ 379690 \\ 3510 \end{bmatrix}$$

> $Megoldas := simplify(MatrixInverse(XA.Transpose(XA)) . XA.Transpose(convert(Y, Matrix)))$:

> $a2 := evalf(Megoldas_{1_1})$; $a1 := evalf(Megoldas_{2_1})$; $a0 := evalf(Megoldas_{3_1})$

$$a2 := 0.4185489151$$

$$a1 := 1.755483413$$

$$a0 := 367.0779199$$

Tehát az y értékeit legjobban közelítő kétváltozós függvény:

> $f := a0 + a1 x1 + a2 x2$

$$f := 367.0779199 + 1.755483413 x1 + 0.4185489151 x2$$

Az eljárás általánosítható többváltozós esetre, ekkor az XA mátrix $(n+1) \times N$ típusú, ahol n a

független változók száma.

Feltételek:

1. N a mérések száma. $N \geq (n+1)$.
2. A független változók értékei lineárisan független rendszert alkossanak, vagyis a független változók értékei ne függjenek egymástól.

Ha valamelyik változótól nem lineárisan függ a függő változó, alkalmazni kell a linearizálás módszerét.

▼ 8. 6. Gyakorlati alkalmazások

6. Példa A vízgazdálkodásnál fontos feladat a várható vízmagasság előrejelzése. A vízmagasság 5-6 paramétertől is függ, most csak a csapadék mennyisége és az esőzés kezdetén fennálló budapesti vízállás befolyását vizsgáljuk. $X1$ a csapadék mennyiségét (mm), $X2$ az esőzés kezdetén a vízállást (cm), Y a tetőző vízállást (cm) tartalmazza, 26 év megfigyelései alapján. Mennyi lesz a maximális vízállás, ha 100 mm eső esik és a kezdeti vízállás 380 cm?

Megoldás

- > $X1 := [58, 52, 133, 179, 98, 72, 72, 43, 62, 67, 64, 33, 57, 62, 54, 48, 86, 74, 95, 44, 53, 77, 46, 123, 26, 62] :$
 $N := \text{nops}(X1) :$
- > $X2 := [405, 450, 350, 285, 330, 400, 550, 480, 450, 610, 380, 460, 425, 560, 420, 620, 390, 350, 570, 710, 700, 580, 700, 560, 370, 430] :$
- > $Y := [590, 660, 780, 770, 710, 640, 670, 520, 660, 690, 500, 460, 610, 710, 620, 660, 620, 590, 740, 730, 720, 720, 640, 805, 510, 673] :$
- > $XA := \text{Matrix}(3, N, [X2, X1], \text{fill} = 1)$

$$XA := \begin{bmatrix} 3 \times 26 \text{ Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran_order} \end{bmatrix}$$

Figyeljük meg, ilyen nagy méretű mátrixot már nem ír ki a Maple elemenként, de az eredményre kétszer rákattintva egy külön ablakban megjelenik a mátrix, elemenként.. **valóban?**

- > $A := XA.\text{Transpose}(XA); B := XA.\text{Transpose}(\text{convert}(Y, \text{Matrix})) ;$
 $M := \text{evalf}(\text{MatrixInverse}(A).B) :$
- > $a0 := M_{3,1}; a1 := M_{2,1}; a2 := M_{1,1}$

$$A := \begin{bmatrix} 6413075 & 849440 & 12535 \\ 849440 & 157786 & 1840 \\ 12535 & 1840 & 26 \end{bmatrix}$$

$$B := \begin{bmatrix} 8269640 \\ 1251091 \\ 16998 \end{bmatrix}$$

$$a0 := 274.8906841$$

$$a1 := 2.348974379$$

$$a2 := 0.4410633712$$

> $f := a0 + a1 x1 + a2 x2$

$f := 274.8906841 + 2.348974379 x1 + 0.4410633712 x2$

A maximális vízállás 100 mm eső és 380 cm kezdeti vízállás esetén:

> $keresett_vizallas := subs(\{x1 = 100, x2 = 380\}, f)$

$keresett_vizallas := 677.3922031$


Ábrázoljuk a mérési eredményeket és a regressziós síkot közös koordináta rendszerben.

> $N := nops(X1) : with(plots) :$

$pontok := pointplot3d([seq([X1_p, X2_p, Y_i], i = 1 .. N)], style = point, symbolsize = 30, symbol = cross, color = red, axes = normal) :$

$x10 := \min(seq(X1_p, i = 1 .. N)) : x1v := \max(seq(X1_p, i = 1 .. N)) :$

$x20 := \min(seq(X2_p, i = 1 .. N)) : x2v := \max(seq(X2_p, i = 1 .. N)) : y0 := \min(seq(Y_p, i = 1 .. N)) : yv := \max(seq(Y_p, i = 1 .. N)) : graf := plot3d(f, x1 = x10 - 1 .. x1v, x2 = x20 - 1 .. x2v, style = surface) : plots[display3d]([pontok, graf], axes = normal, orientation = [60, 60, 0])$



7. Példa A fagyjárás napi forgalma függ attól, hány fok van az adott napon, függ attól, mennyi pénze van az embereknek, illetve függ attól, hogy hétvége van-e vagy sem. Legyen X1 a hőmérséklet Celsius fokban, X2 a kőolaj napi piaci ára (ettől függhet a pénzmennyiség), X3 értéke 1, ha hétvége, 0 ha nem. Y az eladott fagyimennyiséget tartalmazza. Mennyi fagyilaltot adnak el 25 fok 100 dolláros olajár és hétvége esetén?

Megoldás

- > restart; with(LinearAlgebra) :
- > X1 := [20, 12, 32, 15, 20, 22, 34, 20, 16, 23, 22, 24] :
 $N := \text{nops}(X1) : X2 := [100, 98, 102, 103, 95, 89, 99, 104, 105, 100, 101, 103] : X3 := [1, 1, 0, 0, 0, 0, 1, 0, 1, 1, 1, 1] :$
- > Y := [765, 780, 804, 899, 954, 990, 786, 788, 812, 850, 899, 912] : $XA := \text{Matrix}(4, N, [X3, X2, X1], \text{fill} = 1) ; A := XA.\text{Transpose}(XA) ; B := XA.\text{Transpose}(\text{convert}(Y, \text{Matrix}))$

$$XA := \begin{bmatrix} 4 \times 12 \text{ Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran_order} \end{bmatrix}$$

$$A := \begin{bmatrix} 7 & 706 & 151 & 7 \\ 706 & 120015 & 25963 & 1199 \\ 151 & 25963 & 6078 & 260 \\ 7 & 1199 & 260 & 12 \end{bmatrix}$$

$$B := \begin{bmatrix} 5804 \\ 1021046 \\ 221425 \\ 10239 \end{bmatrix}$$

- > $M := \text{evalf}(\text{MatrixInverse}(A).B) : a0 := M_{4,1} ; a1 := M_{3,1} ; a2 := M_{2,1} ; a3 := M_{1,1}$
 $a0 := 1722.169106$
 $a1 := -1.286281520$
 $a2 := -8.185884070$
 $a3 := -39.67444030$

Tehát az Y értékeit legjobban közelítő háromváltozós függvény:

- > $f := a0 + a1 x1 + a2 x2 + a3 x3$
 $f := 1722.169106 - 1.286281520 x1 - 8.185884070 x2 - 39.67444030 x3$

A keresett eladott fagyaltmennyiség a round függvénnyel kerekítve::

- > round(subs({x1 = 25, x2 = 100, x3 = 1}, f))
832

▼ 8. 7. Beépített Maple lehetőségek

A 7. fejezetben említett görbeillesztő interaktív felület a legkisebb négyzetes illesztést is lehetővé tesz. Eszközök->Asszisztens(A)->GörbeIllesztés(C) asszisztens többféle görbeillesztést tesz lehetővé. A pontok megadása után válasszuk a LeastSquares opciót.

- > CurveFitting[Interactive]()

▼ 8. 8. Feladatok

1. Feladat Az alábbi táblázat egy átlagos gépkocsi sebességének (km/h) és az adott sebességhez tartozó féktávolságot (m) mutatja.

$$X = [5.4, 9.6, 14.9, 19.8, 25.6, 29.4, 35.3, 39.7], Y = [5.6, 11.8, 20.4, 39.9, 51.9, 71.9, 90.3, 127.3]$$

Az adatok ábrázolásával, a legkisebb négyzetek módszere alapján határozzuk meg a közelítő függvény típusát és együtthatóit!

2. Feladat Eldorádóban két kormányzójelölt van: P és Q. A választási hadjárat kezdetekor (0. hónap) és ettől kezdve félhavonként közvéleménykutatást tartanak, mely Q népszerűségének növekedést mutatja, az egyes felmérésekkor a lakosság 6.5, 6.6, 7.2, 8.9, 12.1, 17.5, és 22.5 %-a támogatja. Állapítsa meg Q népszerűségi görbét a legkisebb négyzetek módszere alapján! A közelítő függvény $y = ax^3 + b$ legyen! Ha a 4. hónap elején vannak a választások, ki lesz a kormányzó?

3. Feladat Egy házibuliban időnként valaki cigarettára gyújt. A szobában egy mérőberendezés van, amely regisztrálja a levegőben lévő nikotin mennyiségének változását: 1 perc elteltével a levegő nikotintartalma 0.18 %, 2 perc elteltével 0.23%, 5, 10, 20, 50 perc múlva rendre 0.28, 0.3, 0.32, 0.33 %. A legkisebb négyzetek módszere alapján állapítsuk meg a nikotin növekedési tendenciáját az idő függvényében!

4. Feladat A víz forráspontját (t , C^0) a felszínre nehezedő nyomást (p Hgmm) a következő táblázat tartalmazza:

p	700	710	720	730	740	750	760	770	780	790
t	97.71	98.10	98.49	98.88	99.26	99.63	100.00	100.37	100.73	101.09

(A tengerszint feletti magasság növekedésével csökken a légnyomás.) Határozza meg a víz forráspontját 721 Hgmm nyomáson.

5. Feladat A piacon a gyümölcs ára fordítottan arányos a felhozatal mennyiségével. Az elmúlt időben a gyümölcsmennyiség és az árak a következőképpen változtak: Amikor 500 kg almát hoztak, akkor kilójáért 22 forintot (ft) kértek. 1000 kg esetén az egységár 19.50 ft, 1600 kg esetén 18.50 ft, 2000 kg esetén 18.20 ft, 3000 kg esetén 17.90 ft, 4000 kg esetén 17.50 ft. Határozza meg az árfüggvényt a legkisebb négyzetek módszere alapján. ($y = \frac{a}{x} + b$). Mennyi lesz az alma ára 8000 kg felhozatala esetén?

6. Feladat A közgazdászok az árucikkeket olyan szempontból is osztályozzák, hogy mekkora az irántuk mutató kesés. Statisztikailag vizsgálták, miként változik az élelmiszerekre jutó átlagos kiadás az egy főre jutó teljes fogyasztási kiadáshoz képest. (Az árak dollárban vannak megadva.)

Átlagos összes	52.9	76.2	102	143	144	291	309	350	407	439	539	622	919	1295
Élelmiszeri kiadás	33.7	48.1	58.4	69.6	90.3	153	151	135	173	152	172	215	247	409

Határozza meg a legjobban közelítő $y = ax^b$ alakú függvényt!

7. Feladat Az alábbi táblázat a Kootenai folyó vízhozamát mutatja 1931-től, évente, két helyen:

x	27.1	20.9	33.4	77.6	37.0	21.6	17.6	35.1	32.6	26.0	27.6	38.7	27.8
y	19.7	18.0	26.1	44.9	26.1	19.9	15.7	27.6	24.9	23.4	23.1	31.3	23.8

Milyen függvénykapcsolat van a két helyen mért vízhozam között?

▼ 8. 9. Ellenőrző kérdések

1. Fogalmazza meg a legkisebb négyzetek módszerének alapproblémáját!
2. Milyen egyenletrendszer megoldása során jutunk a regressziós egyenes paramétereinek meghatározásához?
3. Mit jelent a linearizálás a legkisebb négyzetek módszere esetén?
4. Többváltozós lineáris regresszió esetén milyen segédmátrix alkalmazható az A együtthatómátrix felírásához?
5. Többváltozós lineáris regresszió esetén milyen mátrixművelettel kapjuk meg a közelítő függvény együtthatóit?